
Modélisation du circuit intégré TL431 
 

 

J’invite le lecteur à consulter le site pour des informations complémentaires.  
 

Page d’accueil du site Internet :    d’autres pdf, sur différents sujets : 
page d'accueil                    liste des PDF  

 

 

Le circuit TL431 est une référence de tension réglable par une tension, utilisé dans 

de nombreux montages. Il m’a paru intéressant d’expliquer son fonctionnement 

interne d’une part, et de faire quelques simulations pour valider le modèle d’autre 

part, en exploitant la datasheet des constructeurs. 

 

 

Introduction 
 

Le TL431 peut être vu comme une diode zener : parcouru par un courant, le composant impose une 

tension constante à ses bornes. Le « coude zener » est réglable par une tension de commande, ce qui fait 

l’intérêt de ce composant. Deux autres avantages du TL431 sont : la faible impédance de sortie 

(l’équivalent de la résistance dynamique de la « diode zener »), et la stabilité en température. 

 

Ce composant comporte 3 broches, nommées 

Anode, Cathode, Référence. On retrouve le 

symbole de la diode zener. 

                            
https://www.onsemi.com/pub/Collateral/TL431-D.PDF 

L’illustration montre le boitier TO92. 

Le constructeur indique 2 façons d’exploiter cette 

référence de tension : suiveur  ou amplificateur : 

suiveur 

 
 

 

VKA = Vref 

amplificateur 

 
 

VKA = 1ref

2

1
ref R I)

R

R
(1 V   

http://www.ti.com/product/TL431/toolssoftware 
 

La tension VKA peut être supérieure à la tension de référence : il y a donc une électronique d’amplification 

dans le TL431. 

 

Consultons plus profondément la 

documentation : 

 

Comme pressenti, le schéma de 

principe montre clairement un 

amplificateur de tension, et une 

tension de référence interne. 

 
schéma de principe. 

Le transistor est en émetteur 

commun, donc inverseur. 
 

schéma complet à transistors 
 

La diode entre Référence et Cathode, indiquée dans le schéma de principe, est en parallèle avec la 

jonction Base Collecteur du transistor NPN qui lui est connecté, et n’apparait pas sur le schéma complet. 

 

Avant de passer aux simulations de ce composant, nous allons repérer, dans le schéma complet, où se 

trouvent la référence de tension et l’amplificateur, en analysant finement les étages à transistors. 

  

http://serge.dusausay.free.fr/
http://serge.dusausay.free.fr/ss/plan2.html
https://www.onsemi.com/pub/Collateral/TL431-D.PDF
http://www.ti.com/product/TL431/toolssoftware

1) Analyse du schéma interne du TL431 
 

1.a) La référence de tension band gap. 
 

Rappelons que dans les conditions de polarisation normale d’un transistor bipolaire, on a la relation 

générale : )
V

V
(expII

T

BE
SC  , où IS est le courant de saturation, VT = kT/q, tension thermodynamique.   

On déduit : VBE = VT ln (IC/IS). 
 

Dans le montage ci-contre, nommé « band gap de Widlar », 

VBE1 - VBE2 s’écrit, par conséquence, 
S1C2

S2C1
TBE

I I

I I
ln VV   

Le terme IS est proportionnel à la surface du transistor. En posant A2, A1 

les aires respectives, on peut écrire :  
1

2

S1

S2

A

A

I

I
 , que l’on pose à r. 

D’autre part, le schéma montre ΔVBE = R3 IC2, d’où IC2 = )
I

 I
(rln 

R

V

C2

C1

3

T
 

 

 
extrait du TL431 

(surfaces supposées) 

Pour finir, on peut admettre que VBE1 = VBE3.  

Le potentiel de sortie s’écrit alors : VO = VBE3 + R2 IC2, soit : VBE3 + )
I

 I
(rln 

R

R
V

C2

C1

3

2
T . 

Remarquons que UR1 = UR2, car soumis tous deux à (VO - VBE). Soit : R1 IC1 = R2 IC2. 

D’où : 
C2

C1

I

 I
= 

1

2

R

R
.  VO s’écrit VBE3 + )

R

 R
(rln 

R

R
V

1

2

3

2
T . 

Une expression de VO est alors VBE3 + KVT. L’intérêt de cette structure est la stabilité en température : en choisissant K, on 

peut avoir la dérive de VBE (qui décroit avec T) compensée par K VT (qui croit avec T). Les 2 lois étant différentes, la stabilité 

n’est que dans une plage limitée de température, usuellement au voisinage de la température ambiante, soit 25 °C. 

Application numérique : 
1

2

R

R
= 

k 2,4

k 7,2
= 3. La surface de Q2 est 1,75, celle de Q1 est 1, soit r = 1,75. 

Il vient VO = VBE3 + 3) x (1,75ln  
800

 7200
 x mV 26  = VBE3 + 0,39 V.  

Si on suppose VBE ≈ 0,7 V (valeur usuelle), cela donne VO ≈ 1,09 V. 

 

1.b) Le schéma interne du TL431 
 

Par IC1 = 3 IC2, nous avons : 
 

IR4 = IC1 + IC2 = 4 IC2. 

Comme R2 IC2 = 0,39 V,  

on déduit : 

IC2 = 0,39 / 7,2 k = 54 µA. 
 

D’où : 

UR4 = 3,28 k x 216 µA = 

0,708 V. 

En supposant VBE ≈ 0,7 V, 

on peut déterminer Vref : 
 

Vref  = VO + UR4 + VBE  

≈ 1,09 + 0,708 + 0,7  

= 2,498 V. 
 

Dans ce cas, Q11 est une diode polarisée en inverse, car soumise à 1,2 (par Q9, Q10) -  2,498  ≈ - 1,3 V.  

L’étude précédente a montré que la référence de 2,5 V n’était 

possible que si un courant circulait dans R4, ce qui suppose le 

transistor Q5 polarisé en mode normal. 

 

Texas Instrument donne les conditions nominales de polarisation, 

au travers cette figure 11 ci-contre : 

 

Il est indiqué que la polarisation est pour IKA = 10 mA. D’ailleurs 

la source de tension doit être ajustée en conséquence pour 

répondre à cette condition. 

 
 

 

En fusionnant le schéma de principe du TL431 et ce schéma de 

polarisation, on arrive au montage ci-contre. 
 

Présenté de la sorte, on reconnait facilement le montage 

amplificateur non inverseur. 
 

En Output (ou K), le potentiel théorique est : 

2,5 (1 + 15 k / 8,25 k) =  7,04 V. 

En négligeant le courant circulant dans les résistances devant IKA = 

10 mA, la tension d’alimentation de ce montage doit être réglée à :  

7,04 + (232 x 10 m) = 9,36 V. 

 

 
polarisation : 

schéma équivalent du montage 

 

Remarquons pour finir que cette façon de représenter une référence interne de 2,5 V 

associé à un A. Op. (dont le signal d’erreur serait ε ≈ 0), est un modèle. 

Dans le TL431, il n’y a pas d’étage d’entrée d’amplificateur opérationnel. 

Le schéma plus exact est donné ci-contre. Rappelons que VO n’existe que si une 

tension est appliquée à VREF, pour faire circuler un courant dans Q5. 

On reconnait une entrée différentielle, tension d’erreur ε. En statique ε  ≈ - 1,4 V. 
 

1.c) L’étage amplificateur 
 

Il nous reste à identifier l’électronique d’amplification. Simplifions dans un premier temps le schéma. 

Observons le TL431 sans contre réaction, avec R = 232 Ω en pull up sur le transistor de sortie.  

Le signal à amplifier est sur l’entrée REF. Il doit être porté par 

une composante continue = 2,5 V pour polariser Q5. Ce 

dernier, monté en émetteur suiveur, ne participe pas à 

l’amplification dynamique. 
 

L’amplification est réalisée en partie par Q4, polarisé par Q3, 

dont la base est à un potentiel constant, et qui agit en source 

de courant constant. Q4 commande le miroir de courant Q7, 

Q8, qui attaque Q9, Q10 montés en Darlington, représentés par 

Q19 ci-contre. 
 

Q6, dont la base est à un potentiel constant, assure un courant 

constant.  

 
 

schéma simplifié 

Fonctionnement : si VREF croit, VB4 croit, IC4 croit, IC8 croit (fléchés descendant sur le schéma), IB19 croit 

car la source Q6 est constante, IC19 croit, VK diminue. 

On a bien un amplificateur inverseur, où le deuxième étage d’amplification est l’émetteur commun Q19 

chargé par 232 Ω. La valeur de l’amplification est directement liée à cette résistance.  

Remarquons que Q4 n’est pas le « classique » émetteur commun sur charge active. 

2) Simulations de validation du schéma du TL431 à 11 transistors 
 

2.a) Polarisation 
 

On reprend la figure 11 de la data sheet pour la simulation. 

Un run .OP montre qu’il faut VDC = 9,4 V pour avoir IKA 

= 10 mA, ce qui est conforme à la prédétermination. 

Dans ce cas, le fichier de sortie indique : 

on a V(K) = 7,0459 V et V(REF) = 2,4927 V. 

De même, VO, référence de tension bande gap de Widlar, 

donne 1,09 V (comme estimé). 

 

 
polarisation 

VDC alim 0 DC=9.4  

Rpu alim K 232  

R1 K REF 15k 

R2 REF 0 8.25k 

X1 K REF 0 TL431  

.OP 

extrait 

de la netliste 
 

 
Après simulation .OP, les potentiels statiques ont été reportés sur le schéma 

 

2.b) Simulation de la référence de tension band gag de Widlar 
 

On se propose de vérifier le comportement en 

température du montage de la référence de 

tension uniquement.  

Pour ce faire, on crée un nouveau schéma en 

prélevant un extrait du TL431. 

On affecte un coefficient de température positif 

aux résistances, par le paramètre TC. 

On place 1,7607 V en base de Q4 et 6,2788 V 

en collecteur de Q4 comme mesuré dans 

l’analyse .OP du montage complet. 

On lance une analyse DC en faisant varier la 

température entre – 60 ° à 125 °. 

 

.param TC=4.1e-3 

Valim K 0 DC=6.2788V 

VDC V 0 DC=1.7607V 

R4 V VO 3.28k 

Q4 K V EQ4 QN 

R5 EQ4 CQ3 4k 

R1 VO CQ1 2.4k TC={TC} 

R2 VO CQ2 7.2k TC={TC} 

Q1 CQ1 CQ1 0 QN 1 

Q2 CQ2 CQ1 EQ2 QN 1.75 

R3 EQ2 0 800 TC={TC} 

Q3 CQ3 CQ2 0 QN 

.OP 

.DC TEMP -60 125 10m 

netliste indépendante 

 

 
Band gap seul  

 

La valeur du coefficient de température TC 

dépend de la technologie et n’est pas mentionnée 

dans les datasheet. Ce coefficient s’applique aux 

résistances intégrées, et aux résistances d’accès 

des semi-conducteurs.  

La valeur de 4,1 e-3 permet d’avoir le but 

recherché : au voisinage de 25 °C, VO conserve 

une valeur de 1,09 V à moins de 3 mV près. 
 

VO fonction de la température 

Cet essai montre la stabilité thermique du montage band gap uniquement.  

           TEMP

-75 -50 -25 0 25 50 75 100 125

V(VO)

1.087V

1.088V

1.089V

1.090V

2.c) Dérive thermique du TL431 à VKA = VREF  
 

 
schéma 

Commentaires : 

La stabilité de la référence de tension en 

fonction de la température est montrée –en 

autre- par la figure 1, par Texas. 

Elle donne l’évolution de la tension de 

référence (valeur nominale 2495 mV à 25 °C) 

selon la température, dans le montage 

« suiveur » (K et REF reliés).  

La dynamique (max – min) est environ 30 mV 

pour 185 °C.  

On retrouve la caractéristique de la référence band gap : une dérive thermique la plus faible possible. Ici, 

il est montré, pour les 3 valeurs de Vref (max, typ, min) une faible variation, au voisinage de 25 °C. 
 

 

On se propose de reproduire cette manipulation 

par la simulation Pspice en température de toute 

l’électronique du TL431. 

On conserve la valeur de TC = 4,1e-3 

 
VDC alim 0 DC=5 

Rpu alim K 232  

X1 K K 0 TL431  

.DC TEMP -60 125 10m 

extrait de la netliste 
la simulation Pspice donne un comportement 

différent que celui donné dans la data sheet 
 

Commentaires : 

On passe par le point [2,496 V ; 25° C]. Le courant IKA est alors 10,78 mA 

Vref va de 2,46 V pour -60 °C, à 2,523 V pour 125 °C, soit + 63 mV pour un accroissement de 185 °C. 

Clairement, le phénomène de stabilisation thermique du TL431 n’est pas retrouvé. 

 

Malgré plusieurs tentatives de différentes valeurs du coefficient de température TC, il n’a pas été possible 

d’avoir un comportement thermique similaire à celle donnée par Texas pour le TL431. 

 

Une explication est la non validité des paramètres intervenant dans le modèle des transistors, ou des 

coefficients de température de résistance non identique selon leur constitution, ou des surfaces de 

transistors non respectées... Faute d’avoir des renseignements sur la technologie, on doit conclure que le 

modèle proposé dans cet article n’est pas valide pour le comportement en température. 

 

Remarquons que Pspice ne connait pas la température ambiante, mais la température de la puce. En 

statique, les 2 sont reliées par la loi d’ohm thermique : Δϴ = RTH P. Ici, la puissance est 2,5 V x 10 mA = 

25 mW. La résistance thermique Jonction-Ambiante est donnée dans la data sheet : par exemple  140 °/W 

pour le boitier TO92. C’est à dire que 25 ° ambiant signifie 25 + 140 x 25m = 28,5 °C au sein de la puce. 

Ce bref calcul montre que l’écart est négligeable. Il n’en serait pas de même en électronique de puissance 

où la température de la puce est bien plus élevée que la température ambiante. 

 

  

           TEMP

-75 -50 -25 0 25 50 75 100 125

V(K)

2.46V

2.48V

2.50V

2.52V

2.45V

2.53V

2.d) Caractéristique de transfert statique 
 

La caractéristique de transfert statique permet d’avoir la caractéristique IAK = f(VAK). 

Sur une diode zener, c’est la caractéristique en inverse qui est intéressante, présentée usuellement, sur le 

3ème quadrant. 
 

 
schéma 

 

Commentaires : 

La  caractéristique est sous forme « inversée » : IKA = f(VKA), 

donc sur le premier quadrant. 

 

On reconnait l’effet zener à 2,5 V. Le coude est très franc. La 

tension est maintenue à 2,5 V sur une très grande dynamique 

de courant. 

 

 

VDC alim 0 DC=9.4 

Rpu alim K 232  

X1 K K 0 TL431  

.DC VDC -25 40 1m 

extrait de la netliste 

 

 

Commentaires   

- L’allure globale est identique 

- Une mesure entre 2 points donne la 

résistance dynamique :  

ΔV/ΔI = 4,2942 mV/50,083 mA  

≈ 0,086 Ω. 

 
    

    

Valeur que l’on peut comparer à celle donné par Texas : 

 
La résistance dynamique est très faible. Néanmoins, elle se dégrade si on exploite le TL431 en 

amplificateur et non en suiveur. On estime qu’elle est multipliée par (1+R1/R2).  

 

Dans le quadrant opposé, c’est la diode D2 qui est passante. Un paramétrage indépendant de ce composant 

a permis de « coller » au mieux avec la caractéristique de Texas. 

 

2.e) Amplification dynamique, réponse harmonique 
 

 
schéma 

 

Commentaires : 

Texas propose ce schéma pour montrer la réponse 

harmonique de l’amplificateur. Ce montage est 

contre-réactionné. 

L’amplificateur est équipé de 232 Ω de pull up. 

Le condensateur de liaison de 9 µF coupe les 

basses fréquences ce qui réduit la bande passante. 

La fréquence de transition est proche de 2 MHz. 
 

           V(K)

-2.0V -1.0V 0V 1.0V 2.0V 3.0V

-I(X1:3)

-100mA

-75mA

-50mA

-25mA

-0mA

25mA

50mA

75mA

100mA

125mA

150mA

Les renseignements donnés par Texas sont suffisants pour 

reproduire cet essai par la simulation. 

 

 

VDC alim 0 DC=9.4 

Rpu alim K 232  

R1 K REF 15k 

R2 REF 0 8.25k 

X1 K REF 0 TL431 

Vin in 0 AC=1 

CIN in REF 9u 

.AC DEC 100 1k 10Meg 
 

extrait de la netliste 

 
 

schéma équivalent 

    

Commentaires   

- ce run AC a été fait sur le point de polarisation du paragraphe 2.a) 

- un run supplémentaire avec CIN >> 9µF a permis d’accéder à l’amplification statique : 760 soit 57,6 dB.  

- on retrouve une allure similaire : par exemple le passage à 30 dB à 100 kHz, la fréquence de transition 

au-delà de 1 MHz. 

- En dynamique, la source continue 2,5 V se comporte comme un court-circuit, et le schéma équivalent 

peut se formaliser par les schémas blocs.  

Le potentiel en e
-
 est, par superposition, la contribution de Vin [à V(K) = 0] + celle de V(K) [à Vin = 0]. 

 
schéma bloc par le 

théorème de superposition 

En première approximation : 

a : pont diviseur de tension formé par CIN, (R1//R2) [V(K) = 0] 

b : pont diviseur de tension formé par R1, (R2//CIN) [Vin = 0]. 

Pour des fréquences moyennes et hautes, « a » tend vers 1 et 

« b » tend vers 0. Il en résulte un schéma bloc simplifié. 

 

 
schéma bloc 

résultant 

C’est donc la réponse harmonique du module de la chaine directe qui est affichée. 

 

2.f) Réponse dynamique en boucle fermée 
 

 
schéma 

Commentaires : 

Cette expérience montre le temps que met le TL431 à 

« clamper » 2,5 V. On peut estimer à 0,5 µs ce temps de réponse. 

C’est plus lent qu’une zener, et est lié au courant de polarisation. 

Texas indique le schéma de test.   
 

On applique un signal carré de 5 V sur le montage de régulation 

par zener. 
 

Vin V 0 PULSE (0 5 1u 10n 10n 5u 10u) 

Rserie V K 220 

X1 K K 0 TL431 

.TRAN 1u 10u 0 1u 

extrait de la netliste 
 

 

Commentaires : 

Dans cette simulation, on voit clairement la stabilisation à 2,5 V, comme attendu. Mais le régime 

transitoire issu de la simulation s’éloigne quelque peu de celle montrée par Texas. 

Néanmoins, en jouant sur le dimensionnement de certains transistors (Q9 et Q10 notamment), il est 

possible d’obtenir une bonne similitude, mais au détriment d’autres paramètres du TL431. 

  

           Frequency

1.0KHz 10KHz 100KHz 1.0MHz 10MHz

DB(V(K))

0

10

20

30

40

50

60

           Time

0s 1.0us 2.0us 3.0us 4.0us 5.0us 6.0us 7.0us 8.0us

V(V) V(K)

0V

1.0V

2.0V

3.0V

4.0V

5.0V

6.0V

2.g) Limite de stabilité sur charge capacitive 
 

 
schémas fournis par Texas.  

 

Commentaires : 

Le TL431 dispose d’un amplificateur interne. En utilisation « zener », il y a une boucle de rétroaction. 

Comme tout système bouclé, on peut craindre une instabilité, surtout sur charge capacitive. 

Texas indique clairement dans quelle configuration de courant IKA, d’une part, et de valeur de 

condensateur CL, d’autre part, le système est stable ou instable.  

Les caractéristiques de l’amplificateur sont liées au courant de polarisation IKA. Cela explique pourquoi 

une situation peut être stable à 10 mA de polarisation, et instable à 50 mA. Par contre, les limites entre les 

zones stables et instables ont un profil impossible à expliquer. On peut remarquer qu’une faible variation 

de CL peut faire basculer stable/instable. 
 

Prenons le cas « C » pour IKA = 20 mA. 

 

D’après la figure 16, pour IKA = 20 mA, nous aurons : 

 

- stable pour CL = 70 nF et 1 µF 

 

- instable pour CL = 0,2 µF et 0,4 µF (zone jaune) 

 

La tension de sortie VKA doit être à 10 V. La valeur de R1 est imposée à 10 kΩ. 

Par VKA = )
R

R
(1 V

2

1
ref  , on aboutit à R2 = R1/3, soit R2 = 3,33 kΩ. 

VBATT = 13 V permet d’avoir IKA = (13 – 10 / 150) = 20 mA, en négligeant le courant dans le pont R1, R2. 
 
 

DC BATT 0 DC=13 

Rpu BATT K 150  

R1 K REF 10k 

R2 REF 0 3.32k 

X1 K REF 0 TL431  

.param C=1u 

CL K 0 {C} IC=10 

.STEP param C list  0.07u 0.2u 0.4u 1u 

.TRAN 0.1u 300u 0 0.1u  

extrait de la netliste 

 
Tension régulée : oscille avec CL = 0,2 µF et 0,4 µF 

Stable à  10 V pour CL = 70 nF et 1 µF 
 

On place une condition initiale pour provoquer une perturbation. Ici, une valeur de tension de sortie. Le 

montage réagit à ce stimulus, à t = 0, puis après quelques dizaines de µs, passe en mode entretenu : 
 

- On a un régime oscillatoire pour CL = 0,2 µF (signal périodique à 121,2 kHz) ainsi qu’avec CL = 0,4 µF 

(oscillation à 91,2 kHz). 
 

- Par contre, pour CL = 70 nF ou 1 µF, la tension VKA se stabilise à 10 V.  

           Time

280us 282us 284us 286us 288us 290us 292us 294us 296us 298us 300us

V(K)

9.6V

9.8V

10.0V

10.2V

10.4V

Explications : 

Utilisons le formalisme des systèmes bouclés, et travaillons en petits signaux pour linéariser. 

Dans un premier temps, on peut identifier la chaine directe qu’est la partie amplification du TL431 

(équipée de Rpull up = 150 Ω). L’analyse AC suivante injecte un signal directement sur la borne REF. En 

affichant V(K), on dispose de la réponse harmonique de l’amplification que l’on pose A(jf). 
 

Identification de la 

|chaine directe A|, avec 

les conditions d’envi-

ronnement de la figure 

17 de Texas. 
 

VDC BATT 0 DC=13  

Rpu BATT K 150  

R1 K REF 10k 

R2 REF 0 3.32k 

X1 K REF 0 TL431 

Vin in 0 AC=1 

CIN in REF 9m 

.AC DEC 100 100 10Meg 

extrait de la netliste 
 

 

En positionnant des pentes multiples de 20 dB/décade, on peut modéliser A(jf) par 
)jf/f)(1jf/f(1

)jf/f(1
A

21

3
0




 

avec, en lisant sur le graphe : A0 = 692, f1 ≈ 9 kHz, f2 ≈ 70 kHz, f3 ≈ 400 kHz, et ce pour des fréquences 

inférieures à 10 MHz. 
 

On explique brièvement cette allure :  

f1 est apportée par C1, associé avec les impédances de Q6, Q8, Q11, Q9. 

f2 et f3 sont créées par C2 placé entre base et collecteur de Q3. 

A noter également les capacités de jonction de tous les semi-conducteurs qui influent. 
 

Dans la simulation qui montre l’instabilité, la sortie est chargée par le condensateur CL qui apporte un 

pôle supplémentaire à une fréquence f4 = 
LC R  2

1


avec R = 150 Ω. 

Le retour est assuré par un pont diviseur R1, R2, dont la valeur numérique est 

0,25, que l’on pose à b. 

Dans cette étude de stabilité sur charge capacitive, il n’y a pas d’entrée. Il en 

résulte le schéma fonctionnel ci-contre : 

 
Système bouclé 

 

Il en résulte une TBO(jf) qui sécrit : 
)jf/f)(1jf/f)(1jf/f(1

)jf/f(1
A 0,25

421

3
0




 

 

 

Le script Matlab suivant permet de vérifier la 

stabilité selon CL : 

 
f1 = 9e3 ; 

f2 = 70e3 ; 

f3 = 400e3 ; 

A0 = 692 ; 

R = 150 ; 

C = 0.004e-6 ; 

b = 0.25 ; 

tau1 = 1/(2*pi*f1) ; 

tau2 = 1/(2*pi*f2) ; 

tau3 = 1/(2*pi*f3) ; 

tau4 = R*C ; 

A1 = TF([1],[tau1 1]) ; 

A2 = TF([1],[tau2 1]) ; 

A3 = TF([tau3 1],[1]) ; 

A4 = TF([1],[tau4 1]) ; 

TBO = A0*A1*A2*A3*A4*b ; 

margin (TBO) ; 

 

En modifiant la ligne « C = ... » dans le script Matlab, on obtient, par la commande margin : 

  

           Frequency

100Hz 1.0KHz 10KHz 100KHz 1.0MHz 10MHz

P(V(K))

-360d

-270d

-180d

SEL>>

DB(V(K))

-20

0

20

40

60

CL = 4 nF : stable 

 
CL = 0,3 µF : instable 

 
CL = 2 µF : stable 

 

CL = 4 nF et 2 µF : le système est à la limite de la stabilité, vu la très faible marge de phase. 

CL = 0,3 µF : le système est franchement instable, avec une marge de phase négative :  -22 °. 

 

Bien que les valeurs numériques ne soient pas exactement celles de l’analyse transitoire de Pspice, on 

retrouve la même conclusion : pour CL de valeur « intermédiaire » le système est instable. 

 

Remarquons qu’en modifiant la tension VBATT, on modifie le courant de polarisation et, si on reste dans le 

domaine instable, on modifie ainsi la fréquence. Ce montage peut agir en VCO. Il faut placer un étage de 

sortie (à haute impédance d’entrée) entre VKA et l’utilisateur, pour ne laisser que CL connecté sur VKA. 

 

Conclusion 
 

Le modèle Pspice du TL431 à base du schéma à transistors n’est pas donné par les fabricants, qu’ils 

soient d’origine (Burr Brown à l’époque) ou seconde source. 

 

Texas Instrument, qui donne pourtant des modèles Pspice de très nombreux circuits intégrés qu’il 

fabrique, reste discret sur ce TL431. Tout comme ON semiconducteur, Texas se risque à donner un 

macromodèle (des composants et des sources de courant), alors, qu’en tant que fondeur, il dispose du 

schéma à transistors...  

 

Notons en passant des erreurs dans la documentation de ON semiconducteur, dans les explications de 

l’instabilité du TL431 sur charge capacitive. 

 

La bibliographie donnée dans les pages suivantes montre que ce modèle du TL431 a intéressé au moins 

3  personnes... Un merci à eux en passant pour leur recherche et la diffusion de leurs travaux. 

 

Mon document apporte un élément de plus à cette enquête... 

Bibliographie 
 

a) Un merci à Eugene Dvoskin, qui sous http://www.audio-perfection.com/ a donné des renseignements 

sur une étude personnelle du TL431. 
 

J’ai exploité dans un premier temps son schéma  

http://www.audio-perfection.com/wp-content/uploads/TL431ED_Model_Schematic.png qui donnait des 

renseignements, indispensables pour les simulations : 

- les paramètres des modèles des transistors et diodes, 

- les coefficients de surface de certains transistors (Q2, Q3, Q10 ci-dessous). 
 

 
 

Par contre, Eugene Dvoskin a placé 10 pF en C1 au lieu de 20 pF contrairement à ce que mentionne 

Texas ou ON Semiconductor. Une explication est donnée en page suivante. 

 

Mais, en consultant d’autres sources (lire pages suivantes), j’ai modifié certaines surfaces de transistor, 

pour mieux coller à la documentation de Texas. Notons en passant, que ON Semiconductor donne des 

renseignements un peu différents (et avec des erreurs) dans leur datasheet. 

  

http://www.audio-perfection.com/
http://www.audio-perfection.com/wp-content/uploads/TL431ED_Model_Schematic.png

b) Une mention particulière pour Ken Shirriff, qui a 

« déshabillé » le circuit intégré pour remonter au 

schéma interne.  

Il donne des renseignements sur son blog. 
 

De dépassiver un circuit pour observer au microscope la 

puce n’est pas donné à tout le monde... 
 

Néanmoins, Ken Shirriff a récupéré des photos de la 

puce du TL431 par le site « Zeptobars ». Cela coïncide 

avec ses propres observations faites avec son 

microscope.  
 

Les surfaces des transistors ont donc été évaluées, ainsi 

que les résistances par leur dimension physique. 
 

Ken Shirriff a analysé cette photo, et donne des 

explications très pédagogiques. 
 

Chapeau bas pour ce travail 

 

 
 

source : Zeptobars 

http://www.righto.com/2014/05/reverse-engineering-tl431-most-common.html 

 
 

Ken Shirriff attire l’attention sur le fait que le transistor de sortie est conçu de façon différente : c’est un 

montage « latéral » et non enterré comme les autres. De plus, il estime que R = 100 Ω, en observant les 

dimensions physiques du die. Ce qui semble contradictoire avec les renseignements donnés par 

constructeurs du TL431. Autre contradiction, signalée par Ken Shirriff lui-même : les valeurs identiques 

de R2, R3 dans la source de Widlar. De plus, 8 transistors sur 11 ont des surfaces différentes, ce qui, 

techniquement, peut être possible, mais est vraiment contradictoire avec Eugene Dvoskin et Christophe 

Basso (lire page suivante). Notamment le miroir de courant (Q8, Q9 sur ce schéma) dimensionné pour que 

la sortie soit 2 fois plus faible que l’entrée. 

Une bizarrerie : C2 ne serait pas constitué comme C1, mais serait la capacité d’une diode en inverse. Sauf 

que dans ce cas, sa valeur est directement liée à la polarisation... Cela peut expliquer une valeur médiane 

de 10 pF donnée par Eugene Dvoskin. Et une réponse harmonique liée à la polarisation. 

Le layout montre bien D, la diode entre Reference et Cathode, mais D n’apparait pas sur le schéma. De 

plus, le layout montre que D2 est de taille plus large que D1. 
 

Une coquille de schéma : liaison Base Collecteur de Q2 en trop.  

http://www.righto.com/2014/05/reverse-engineering-tl431-most-common.html
http://static.righto.com/images/TL431/tl431_labeled.png
http://static.righto.com/images/TL431/tl431_schematic.png

c) Mes navigations sur internet m’ont amené à apprécier : 

 

https://cbasso.pagesperso-orange.fr/Downloads/Papers/The%20TL431%20in%20loop%20control.pdf 

cité également par Ken Shirriff . 
 

Christophe Basso explique le fonctionnement du TL431, au travers des transistors discrets (2N 3906 

2N 3904), avec, par conséquence, des surfaces qui ne sont pas celles données par Eugene Dvoskin ni 

Ken Shirriff. Bien entendu, les valeurs numériques des résistances sont adaptées à ces transistors. A 

noter que la valeur de C2 est également différente que celle donnée par Texas, ou ON Semiconductor. 

L’intérêt de ce site est d’avoir des résultats de simulation, comme ici la polarisation : 
 

 
 

Sur ce schéma, ou plutôt cette étude, on a la source bandgap de Wildar avec un rapport 2 dans les surfaces 

de transistors, et un rapport 3 dans les courants (par R2 = R3/3). 

Christophe Basso donne un renseignement sur la diode D1 : elle n’intervient que si le TL431 est utilisé 

en inverse, c’est à dire diode D1 en direct. Il précise zener de 36 V et un coefficient N = 1,3. 

 

  

https://cbasso.pagesperso-orange.fr/Downloads/Papers/The%20TL431%20in%20loop%20control.pdf


Comprendre l’électronique par la simulation, par S. Dusausay              article 79              Page  14 / 14                       2024/2025 

 

Annexe : netlist 

 

La netliste ci-dessous permet de refaire les simulations : il suffit de décommenter les lignes concernées 

par chaque paragraphe de ce document : 
 

TL431 

* fichier TL431.cir 

.model QN npn (BF=140 CJE=1p Cjc=2p 

RB=40 VAF=80 VAR=50 KF=3.2e-16 AF=1) 

.model QP pnp (BF=60 CJE=1p Cjc=3p RB=80 

VAF=70 VAR=40) 

.model D D (RS=5 CJO=4p) 

.model D1 D (RS=2 CJO=8p)  

.model D2 D (RS=3.5 CJO=4p N=1.3) 

 

* 2a) polarisation 

VDC alim 0 DC=9.4 

Rpu alim K 232  

R1 K REF 15k 

R2 REF 0 8.25k 

X1 K REF 0 TL431 

.OP 

 

*2c) derive thermique VKA = VREF 

*VDC alim 0 DC=5 

*Rpu alim K 232  

*X1 K K 0 TL431  

*.DC TEMP -60 125 10m 

*.OP 

 

* 2d) carac de transfert VKA = VREF 

*VDC alim 0 DC=9.4 

*Rpu alim K 232  

*X1 K K 0 TL431 

*.DC VDC -25 40 1m 

 

* 2e) ampli AC 

*VDC alim 0 DC=9.4 

*Rpu alim K 232  

*R1 K REF 15k 

*R2 REF 0 8.25k 

*X1 K REF 0 TL431 

*Vin in 0 AC=1 

*CIN in REF 9u 

*.OP 

*.AC DEC 100 1k 10Meg 

 

* 2f) reponse dynamique BF 

*Vin V 0 PULSE (0 5 1u 10n 10n 5u 10u) 

*Rserie V K 220 

*X1 K K 0 TL431 

*.TRAN 1u 10u 0 1u  

 

* 2g) stabilité sur CL 

*VDC BATT 0 DC=13 

*Rpu BATT K 150  

*R1 K REF 10k 

*R2 REF 0 3.32k 

*X1 K REF 0 TL431 

*.param C=1u 

*CL K 0 {C} IC=10 

*.STEP param C list 0.07u 0.2u 0.4u 1u 

*.TRAN 0.1u 300u 0u 0.1u 

 

* TBO du montage 2g  

*VDC BATT 0 DC=13  

*Rpu BATT K 150  

*R1 K REF 10k 

*R2 REF 0 3.32k 

*X1 K REF 0 TL431 

*Vin in 0 AC=1 

*CIN in REF 9m 

*.AC DEC 100 100 10Meg 

 

.param TC=4.1e-3 

 

.SUBCKT TL431 K Ref A 

D Ref K D 

Q5 K Ref EQ5 QN 

R4 EQ5 VO 3.28k TC={TC} 

R1 VO CQ1 2.4k TC={TC} 

R2 VO CQ2 7.2k TC={TC} 

Q1 CQ1 CQ1 A QN  

Q2 CQ2 CQ1 EQ2 QN 1.75  

R3 EQ2 A 800 TC={TC} 

Q3 CQ3 CQ2 A QN 

C2 CQ3 CQ2 20p  

Q4 CQ7 EQ5 EQ4 QN 

R5 EQ4 CQ3 4k TC={TC} 

R7 K EQ7 800 TC={TC} 

Q7 CQ7 CQ7 EQ7 QP 1.5  

R8 K EQ8 800 TC={TC} 

Q8 CQ8 CQ7 EQ8 QP 

Q11 CQ8 CQ8 Ref QN 

R6 CQ1 BQ6 1k TC={TC} 

Q6 CQ8 BQ6 A QN .5  

Q9 K CQ8 EQ9 QN 2.5 

C1 CQ8 K 8p  

D1 A CQ8 D1 

R9 EQ9 BQ10 150 TC={TC} 

Q10 K BQ10 A QN 10  

R10 BQ10 A 10k TC={TC} 

D2 A K D2 

.ENDS 

 

.PROBE 

.END 

